Formalising Proj Construction in Lean

Jujian Zhang
Department of Mathematics, Imperial College London

Abstract

Many object of interest in mathematics can be studied both analytically and algebraically, while at the same time, it is known that analytic geometry and algebraic geometry generally does not behave the same. However, the famous GAGA theorem asserts that for projective varieties, analytic and algebraic geometries are closely related; proof of the Fermat last theorem, for example, use this technique to transport between the two worlds [13]. A crucial step of proving GAGA is to calculate cohomology of projective spaces [12, 8], thus I formalise Proj construction for any $\mathbb{N}$-graded R-algebra A as a starting point to the GAGA theorem and projective n-space is constructed as $\operatorname{Proj} A[X_0, \ldots, X_n]$. This would the first family of non-affine schemes formalised in any theorem prover.

2012 ACM Subject Classification Theory of computation → Logic and verification; Mathematics of computing → Topology

Keywords and phrases Lean, formalisation, algebraic geometry, scheme, Proj construction, projective geometry

Digital Object Identifier 10.4230/LIPIcs.CVIT.2016.23

Supplementary Material *Software (Source Code)*: `https://github.com/leanprover-community/mathlib/pull/18138/commits/00c4b0918a2c7a8b62291581b0e1eddf2357b5be`

Funding *Jujian Zhang*: Schrödinger Scholarship Scheme

Acknowledgements I want to thank Eric Wieser for his contribution and suggestion in formalising homogeneous ideal and homogeneous localisation; Andrew Yang and Junyan Xu for their review and suggestion of my often badly written code; Kevin Buzzard for suggesting this project and all the contributors to mathlib for otherwise this project would not have been possible.

1 Introduction

Algebraic geometry concerns polynomials and analytic geometry concerns holomorphic functions. Though all polynomials are holomorphic, the converse is not true; thus many analytic objects are not algebraic, for example $\{x \in \mathbb{C} \mid \sin(x) = 0\}$ can not be defined as zero locus of a polynomial in one variable, for polynomials always have only finite number of zeros. However, for projective varieties over $\mathbb{C}$, the categories of algebraic and analytic coherent sheaves are equivalent; an almost immediate consequence for this statement is that all closed analytic subset of projective n-space $\mathbb{P}_n$ is also algebraic [13, 4]. A crucial step of proving the above statement is to consider cohomology of projective n-space $\mathbb{P}_n$ [12].

While one can define $\mathbb{P}_n$ over $\mathbb{C}$ without consideration of other projective varieties, it would be more fruitful to formalise Proj construction as a **scheme** and recover $\mathbb{P}_n$ as $\operatorname{Proj} \mathbb{C}[X_0, \ldots, X_n]$, since, among other reasons, by considering different base rings, one obtain different projective varieties, for example, for any homogeneous polynomials $f_1, \ldots, f_k$, $\operatorname{Proj}\left(\frac{\mathbb{C}[X_0,\ldots,X_n]}{(f_1,\ldots,f_k)}\right)$ defines a projective hypersurface over $\mathbb{C}$.

In this paper I describe a formal construction of $\operatorname{Proj} A$ in the Lean3 [7] theorem prover which closely follows [9, Chapter II]. The formal construction use various results from the Lean mathematical library mathlib, most notably the graded algebra and Spec construction; this project has been partly accepted into mathlib already while the remainder is still undergoing a review process. The code discussed in this paper can be found on GitHub[1]. I have freely used axiom of choice and law of excluded middle throughout the project since the rest of mathlib freely use classical reasoning as well; consequently, the final construction is not computable.

As previously mentioned, Proj construction heavily depends on graded algebra and Spec construction. A detailed description for graded algebra in Lean and mathlib as well as comparison with graded algebra in other theorem provers can be found in [16], for my purpose, I have chosen to use internal grading for any graded ring $A \cong \bigoplus \mathcal{A}_i$ so that the result of construction is about homogeneous prime ideals of A directly instead of $\bigoplus_i \mathcal{A}_i$. The earliest Spec construction in Lean and any other theorem prover can be found in [2] where the construction followed a "sheaf-on-a-basis" approach from [14, Section 01HR], however it differs significantly from the Spec construction currently found in mathlib where the construction follows [9, Chapter II] were used; for this reason, I have also chosen to follow the definition in [9, Chapter II] while hand-waving part (which is almost the whole proof) was made to be explicit. Some other theorem provers also have or partially have Spec construction: in Isabelle/HOL, Spec is formalised by using locales and rewriting topology and ring theory part of existing library in [1], however the category of scheme is yet to be formalized; an early formalisation of Spec in Coq can be found in [3] and a definition of scheme in general can be found in its `UniMath` library; due to homotopy type theory, only a partial formalisation of Spec construction can be found in [11]. Though some theorem provers have definition of a general scheme, I could not find any concrete construction of a scheme other that Spec of a ring[2].

After explaining the mathematical details involved in Proj construction in Section 2, Lean code will be provided and explained in Section 3. For typographical reasons, some code of formalisation will be omitted and marked as `sorry` or `_` and some code presented in this paper is pseudocode that closely resembles the actual code but with, for example, notations and names altered to make it more readable and presentable.

2 Mathematical details

In this section, familiarity of basic ring theory, topology and category theory will be assumed. In Sections 2.1 and 2.2, definition of a scheme is explained in detail; Spec construction will also be briefly explained in order to fix the mathematical approach used in mathlib. Then by following definition of a scheme step by step, Proj construction will be explained in Section 2.3.

2.1 Sheaves and Locally Ringed Spaces

Let X be a topological space and $\mathfrak{Opens}(X)$ be the category of open subsets of X.

▶ **Definition 1** (Presheaves [10]). *Let C be a category, a C-valued presheaf $\mathcal{F}$ on X is a functor $\mathfrak{Opens}(X)^{\mathsf{op}} \Longrightarrow C$. Morphisms between C-valued functor $\mathcal{F}, \mathcal{G}$ are natural transformation. The category thus formed is denoted as $\mathfrak{PSh}(X, C)$.*

In this paper, the category of interest is category of presheaves of rings $\mathfrak{Psh}(X, \mathfrak{Ring})$. More explicitly, a presheaf of ring $\mathcal{F}$ assigns each open subset $U \subseteq X$ with a ring $\mathcal{F}(U)$ called sections on U and for any open subsets $U \subseteq V \subseteq X$ a ring homomorphism

---
1 url: `https://github.com/leanprover-community/mathlib/pull/18138/`
2 In this paper, all rings are assumed to be unital and commutative.

$\mathcal{F}(V) \to \mathcal{F}(U)$ often denoted as res_U^V or simply with a vertical bar $s\,|_U$ (a section s on V restricted to U). Examples of presheaf of rings are abundant: considering open subsets of $\mathbb{C}$, $U \mapsto \{$(continuous, holomorphic) functions on $U\}$ with the natural restriction map defines presheaf of rings. In these examples, compatible sections on different open subsets can be glued together to form bigger sections on union of the said open subsets; this property can be generalized to arbitrary category:

▶ **Definition 2** (Sheaves [10, 14]). *A presheaf $\mathcal{F} \in \mathfrak{Psh}(X, C)$ is said to be a sheaf if for any open covering of open set $U = \bigcup_i U_i \subseteq X$, the following diagram is an equalizer*

$$\mathcal{F}(U) \xrightarrow{(\mathrm{res}_{U_i}^U)} \prod_i \mathcal{F}(U_i) \overset{\left(\mathrm{res}_{U_i \cap U_j}^{U_i}\right)}{\underset{\left(\mathrm{res}_{U_i \cap U_j}^{U_j}\right)}{\rightrightarrows}} \prod_{i,j} F(U_i \cap U_j).$$

The category of sheaves $\mathfrak{Sh}(X, C)$ is the full subcategory of the category of presheaves.

▶ **Definition 3** (Locally Ringed Space [14, 9]). *If $\mathcal{O}_X$ is a sheaf on X, then the pair $(X, \mathcal{O}_X)$ is called a ringed space; a morphism between two ringed space $(X, \mathcal{O}_X)$ and $(Y, \mathcal{O}_Y)$ is a pair (f, ϕ) such that $f : X \to Y$ is continuous and $\phi : \mathcal{O}_Y \to f_*\mathcal{O}_X$ is a morphism of sheaves where $f_*\mathcal{O}_X \in \mathfrak{Sh}(Y)$ assigns $V \subseteq Y$ to $\mathcal{O}_X(f^{-1}(V))$. A locally ringed space $(X, \mathcal{O}_X)$ is ringed space such that for any $x \in X$, its stalk $\mathcal{O}_{X,x}$ is a local ring where $\mathcal{O}_{X,x} = \mathrm{colim}_{x \in U \in \mathfrak{Opens} X}\, \mathcal{O}_X(U)$; a morphism between two locally ringed spaces $(X, \mathcal{O}_X)$ and $(Y, \mathcal{O}_Y)$ is a morphism (f, ϕ) of ringed space such that for any $x \in X$ the ring homomorphism induced on stalk $\phi_x : \mathcal{O}_{Y,f(x)} \to \mathcal{O}_{X,x}$ is local.*

Following from the previous definitions, if $\mathcal{O}_X$ is a presheaf and $U \subseteq X$ is an open subset, then there is a presheaf $\mathcal{O}_X\,|_U$ on U by assigning every open subset V of U to $\mathcal{O}_X(V)$, this is called restricting a presheaf; sheaves, ringed spaces and locally ringed spaces can also be similarly restricted.

2.2 Definition of Affine Scheme and Scheme

Spec construction

Let R be a ring and $\operatorname{Spec} R$ denote the set of prime ideals of R. Then for any subset $s \subseteq R$, its zero locus is defined as $\{\mathfrak{p} \mid s \subseteq \mathfrak{p}\}$. These zero loci can be considered as closed subsets of $\operatorname{Spec} R$, the topology thus formed is called the Zariski topology. Then a sheaf of rings on $\operatorname{Spec} R$ can be defined by assign $U \subseteq \operatorname{Spec} R$ to the ring

$$\left\{ s : \prod_{x \in U} R_x \mid s \text{ is locally a fraction} \right\},$$

where s is locally a fraction if and only if for any prime ideal $x \in U$, there is always an open subset $x \in V \subseteq U$ and $a, b \in R$ such that for any prime ideal $y \in V$, $b \notin y$ and $s(y) = \frac{a}{b}$. This sheaf $\mathcal{O}$ is called the structure sheaf of $\operatorname{Spec} R$. $(\operatorname{Spec} R, \mathcal{O})$ is a locally ringed space because for any prime ideal $x \subseteq R$, $\mathcal{O}_x \cong A_x$ [9].

Scheme

▶ **Definition 4** (Scheme). *A locally ringed space $(X, \mathcal{O}_X)$ is said to be a scheme if for any $x \in X$, there is always some ring R and some open subset $x \in U \subseteq X$ such that $(U, \mathcal{O}_X\,|_U) \cong (\operatorname{Spec} R, \mathcal{O}_{\operatorname{Spec} R})$ as locally ringed spaces. The category of schemes is the full subcategory of locally ringed spaces.*

Thus to construct a scheme, one needs the following things:

- a topological space X;
- a presheaf $\mathcal{O}$;
- a proof that $\mathcal{O}$ satisfies the sheaf condition;
- a proof that all stalks are local;
- an open covering $\{U_i\}$ of X;
- a collection of rings $\{R_i\}$ and isomorphism $(U_i, \mathcal{O}_X|_{U_i}) \cong (\operatorname{Spec} R_i, \mathcal{O}_{\operatorname{Spec} R})$.

In Section 2.3, Proj construction will be described following the steps above.

2.3 Proj Construction

Throughout this section, R will denote a ring and A an $\mathbb{N}$-graded R-algebra, in order to keep notations the same as Section 3, the grading of A will be written as $\mathcal{A}$, i.e. $A \cong \bigoplus_{i\in\mathbb{N}} \mathcal{A}_i$ as R-algebras.

Topology

▶ **Definition 5** (Proj $\mathcal{A}$ as a set)**.** *Proj $\mathcal{A}$ is defined to be $\{\mathfrak{p} \in \operatorname{Spec} A \mid \mathfrak{p}$ is homogeneous and relevant$\}$, where*

- *an ideal $\mathfrak{p} \subseteq A$ is said to be homogeneous if for any $a \in \mathfrak{p}$ and $i \in \mathbb{N}$, a_i is in $\mathfrak{p}$ as well where $a_i \in \mathcal{A}_i$ is the i-th projection of a with respect to grading $\mathcal{A}$;*
- *an ideal $\mathfrak{p} \subseteq A$ is said to be relevant if $\bigoplus_{i=1}^{\infty} \mathcal{A}_i \nsubseteq \mathfrak{p}$.*

Similar to Spec construction in Section 2.2, there is a topology on Proj $\mathcal{A}$ whose close sets are exactly the zero loci where for any $s \subseteq A$, zero locus of s is $\{\mathfrak{p} \in \operatorname{Proj} \mathcal{A} \mid\subseteq s\mathfrak{p}\}$; this topology is also called the Zariski topology. For any $a \in A$, $D(a)$ denotes the set $\{x \in \operatorname{Proj} \mathcal{A} \mid a \notin x\}$.

▶ **Theorem 6.** *For any $a \in A$, $D(a)$ is open in Zariski topology and $\{D(a) \mid a \in A\}$ forms a basis of the Zariski topology [14].*

Structure sheaf

Let $U \subseteq \operatorname{Proj} \mathcal{A}$ be an open subset, the sections on U are defined to be

$$\mathcal{O}(U) = \left\{ s \in \prod_{x\in U} A_x^0 \mid s \text{ is locally a homogeneous fraction} \right\},$$

where $A_{\mathfrak{p}}^0$ denotes the homogeneous localization of A at a homogeneous prime ideal $\mathfrak{p}$, i.e. the subring of $A_{\mathfrak{p}}$ of elements of degree zero and s is said to be locally a homogeneous fraction if for any $x \in U$, there is some open subset $x \in V \subseteq U$, $i \in \mathbb{N}$ and $a, b \in \mathcal{A}_i$ such that for all $y \in V$, $s(y) = \frac{a}{b}$. Equipped with the natural restriction map, $\mathcal{O}$ defined in this way forms a presheaf; sheaf condition of $\mathcal{O}$ is checked in the category of sets where it follows from that being locally a homogeneous fraction is a local predicate and local predicates define subsheaves. This sheaf is called the structure sheaf of Proj $\mathcal{A}$, also written as $\mathcal{O}_{\operatorname{Proj} A}$

Locally ringed space

▶ **Theorem 7.** *The stalk of $(\operatorname{Proj} \mathcal{A}, \mathcal{O})$ at a homogeneous prime relevant ideal $\mathfrak{p}$ is isomorphic to $A_{\mathfrak{p}}^0$.*

Proof. Let $U \ni \mathfrak{p}$ be an open subset of $\operatorname{Proj}\mathcal{A}$, then a ring homomorphism $\mathcal{O}(U) \to A^0_{\mathfrak{p}}$ can be defined by evaluation at $\mathfrak{p}$, i.e. Since $\mathcal{O}_{\operatorname{Proj}\mathcal{A},\mathfrak{p}} = \operatorname{colim}_{\mathfrak{p}\in U}\mathcal{O}(U)$, a ring homomorphism $f : \mathcal{O}_{\operatorname{Proj}\mathcal{A},\mathfrak{p}} \to A^0_f$ is obtained by universal property of colimit. To check that f is an isomorphism, it is sufficient to check bijectivity:

- Let $z_1 = \langle s_1, U_1\rangle, z_2 = \langle s_2, U_2\rangle \in \mathcal{O}_{\operatorname{Proj}\mathcal{A},\mathfrak{p}}$ be such that $f(z_1) = f(z_2) \overset{\text{def}}{\iff} s_1(\mathfrak{p}) = s_2(\mathfrak{p})$, then by definition of structure sheaf, there is some open subset $\mathfrak{p} \in V \subseteq U_1 \cap U_2$ such that s_1 and s_2 are both constant on V. Since s_1, s_2 restrict to the same section on V, $z_1 = z_2$ hence proving injectivity.
- There is a function $A^0_{\mathfrak{p}} \to \mathcal{O}_{\operatorname{Proj}\mathcal{A},\mathfrak{p}}$ defined by $\frac{a}{b} \mapsto \langle D(b), x \mapsto \frac{a}{b}\rangle$, this function is in fact a right inverse to f.

◀

Since $A^0_{\mathfrak{p}}$ is a local ring for any homogeneous prime ideal $\mathfrak{p}$, it can be concluded that $(\operatorname{Proj}\mathcal{A}, \mathcal{O}_{\operatorname{Proj}\mathcal{A}})$ is a locally ringed space.

Affine cover

▶ **Lemma 8.** *For any $x \in \operatorname{Proj}\mathcal{A}$, there is some $0 < m \in \mathbb{N}$ and $f \in \mathcal{A}_m$, such that $x \in D(f) \overset{\text{def}}{\iff} f \notin x$.*

Proof. Let $x \in \operatorname{Proj}\mathcal{A}$, by construction, $\bigoplus_{i=1}^{\infty}\mathcal{A}_i \not\subseteq x$. Thus there is some $f = f_1 + f_2 + \cdots \notin x$, then at least one $f_i \notin x$ for otherwise $f \in x$. ◀

Thus, to construct an affine cover, it is sufficient to prove that for all $0 < m \in \mathbb{N}$ and homogeneous element $f \in \mathcal{A}_m$, $(D(f), \mathcal{O}_{\operatorname{Proj}\mathcal{A}}|_{D(f)}) \cong (\operatorname{Spec} A^0_f, \mathcal{O}_{\operatorname{Spec} A^0_f})$ where A^0_f is the subring of the localised ring A_f consisted of elements of degree zero. Now fix these notations, an isomorphism between locally ringed space can be constructed as a pair (ϕ, α) where ϕ is a homeomorphism between topological space $D(f) \cong \operatorname{Spec} A^0_f$ and α an isomorphism between $\phi_*(\mathcal{O}_{\operatorname{Proj}\mathcal{A}}|_{D(f)}) \cong \mathcal{O}_{\operatorname{Spec} A^0_f}$.

▶ **Theorem 9.** $D(f) \cong \operatorname{Spec} A^0_f$ *are homeomorphic as topological spaces.*

The following proofs are an expansion of [9, II.2.5] while drawing ideas from [15, II.4.5].

Proof. Define $\phi : D(f) \to \operatorname{Spec} A^0_f$ by $\mathfrak{p} \mapsto \operatorname{span}\left\{\frac{g}{1} \mid g \in \mathfrak{p}\right\} \cap A^0_f$; by clearing denominators, one can show that $\phi(\mathfrak{p}) = \operatorname{span}\left\{\frac{g}{f^i} | g \in \mathfrak{p} \cap A_{mi}\right\}$. For ϕ to be well-defined, the following can be checked:

- $1 \notin \phi(\mathfrak{p})$: for otherwise $1 = \sum_i \frac{a_i}{f^{n_i}}\frac{g_i}{1}$, by multiplying a suitable power of f, $\frac{f^N}{1} = \frac{\sum_i a_i g_i f^{k_i}}{1}$ for some N; by definition of localisation, $f^M f^N = f^M \sum_i a_i g_i f^{k_i}$ for some M, since the right hand side is in $\mathfrak{p}$, the left hand side is in $\mathfrak{p}$ too, implying $f \in \mathfrak{p}$. Contradiction.
- If $x_1 x_2 \in= \phi(\mathfrak{p})$, then either $x_1 \in \phi(\mathfrak{p})$ or $x_2 \in \phi(\mathfrak{p})$: write $x_1 = \frac{a_1}{f^{n_1}}$ and $x_2 = \frac{a_2}{f^{n_2}}$, then $\frac{a_1 a_2}{f^{n_1+n_2}} \in \operatorname{span}\left\{\frac{g}{1} | g \in \mathfrak{p}\right\}$, so write $\frac{a_1 a_2}{f^{n_1+n_2}} = \sum_i \frac{c_i}{f^{n_i}}\frac{g_i}{1}$, by multiplying a suitable power of f, we get $\frac{a_1 a_2 f^N}{1} = \frac{\sum_i c_i g_i f^{k_i}}{1}$ for some N, then by definition of localisation, $a_1 a_2 f^N f^M = f^M \sum_i c_i g_i f^{k_i}$ for some M, since right handside is in $\mathfrak{p}$ and $f \notin \mathfrak{p}$, either $a_1 \in \mathfrak{p}$ or $a_2 \in \mathfrak{p}$.
- ϕ is continuous: since $\operatorname{Spec} A^0_f$ also has a topological basis of basic open sets, it suffices to check that preimages of basic open sets are open. Take $\frac{a}{f^n} \in A^0_f$, then $\phi^{-1}\left(D\left(\frac{a}{f^n}\right)\right) = D(f) \cap D(a)$.

- $D(f) \cap D(a) \subseteq \phi^{-1}\left(D\left(\frac{a}{f^n}\right)\right)$ because if $y \in D(f) \cap D(a)$ and $\frac{a}{f^n} \in \phi(y)$, i.e. $\frac{a}{f^n} = \sum_i \frac{c_i}{f^{n_i}} \frac{g_i}{1}$, then by multiplying suitable powers of f, $\frac{af^N}{1} = \frac{\sum_i c_i g_i f^{m_i}}{1}$ for some N, by definition of localisation, $af^N f^M = \sum_i c_i g_i f^{m_i}$ for some M implying that $a \in y$. Contradiction.
- On the other hand, if $\phi(y) \in D\left(\frac{a}{f^n}\right)$ and $a \in y$, then $\frac{a}{1} \in h(y)$, contradiction because $\frac{a}{f^n} = \frac{a}{1}\frac{1}{f^n} \in \phi(y)$.

For the other direction, define $\psi : \operatorname{Spec} A_f^0 \to D(f)$ to be $x \mapsto \left\{a \mid \text{for all } i \in \mathbb{N}, \frac{a_i^m}{f^i} \in x\right\}$. The following can be checked:

- $0 \in \psi(x)$ for obvious reason.
- if $a, b \in \psi(x)$, then $a + b \in \psi(x)$: since x is prime, it is sufficient to show $\left(\frac{(a_i+b_i)^m}{f^i}\right)^2 = \sum_{j=0}^{2m} \binom{2m}{j} \frac{a_i^j b_i^{2m-j}}{f^{2i}} \in x$. if $m \le j$, write $\frac{a_i^j b_i^{2m-j}}{f^{2i}} = \boxed{\frac{a_i^m}{f^i}} \frac{a_i^{j-m} b_i^{2m-j}}{f^i}$, otherwise, write $\frac{a_i^j b_i^{2m-j}}{f^{2i}} = \boxed{\frac{b_i^m}{f^i}} \frac{a_i^j b_i^{m-j}}{f^i}$. By assumption, the boxed parts are both in x, thus $\left(\frac{(a_i+b_i)^m}{f^i}\right)^2$ is also in x.
- if $a, b \in A$ and $b \in \psi(x)$, then $ab \in \psi(x)$: inducting on a, one obtain that
  - if $a = 0$, then $(ab)_i = 0$;
  - if $a \in \mathcal{A}_n$ and $n \le i$ then $(ab)_i = ab_{n-i}$;
  - if the result hold for a, a', then $(a + a')b = ab + a'b \in \psi(x)$.
- $\psi(x)$ is homogeneous: if $a \in \psi(x)$ then for any $i \in \mathbb{N}$, $a_i \in \psi(x)$, because $(a_i)_j = a_i$ or 0 for all natural number j.
- $\psi(x)$ is prime: for a homogeneous ideal, prime condition is equivalent to being homogeneously prime, i.e. $\mathfrak{p}$ is prime if and only if $1 \notin \mathfrak{p}$ and for any $a \in A_i$ and $b \in A_j$, $ab \in \mathfrak{p}$ implies $a \in \mathfrak{p}$ or $b \in \mathfrak{p}$. $1 \notin \psi(x)$ for the first projection of 1 is 1 which is not x. Suppose $a \in A_i$ and $b \in A_j$, suppose $a, b \notin \psi(x)$ then $\frac{a_n^m}{f^n} \notin x$ for some $n \in \mathbb{N}$ and $\frac{b_k^m}{f^k} \notin x$ for some $k \in \mathbb{N}$. Then $n = i$ for otherwise $0 \notin x$ and similarly $k = j$. So $\frac{(ab)_{i+j}^m}{f^{i+j}} = \frac{a_i^m}{f^i}\frac{b_j^m}{f^j} \notin x$.
- $\psi(x)$ is relevant: for otherwise $\bigoplus_{1 \le i} \mathcal{A}_i \subseteq \psi(x)$ then $f \notin \psi(x)$, for otherwise $1 = \frac{f_m^m}{f^m} \in x$, a contradiction; however $f \in \bigoplus_{1 \le i} \mathcal{A}_i$, since $f_0 = 0$.

ψ being continuous depends on that ψ and ϕ are inverses to each other:

- $\phi \circ \psi = 1$:
  - $\phi(\psi(x)) \subseteq x$: if $z \in \phi(\psi(x))$ then $z \in \operatorname{span}\left\{\frac{c}{f^i} | c \in g(x) \cap A_{mi}\right\}$. So z can be written as $z = \sum_i \frac{a_i}{f^{n_i}} \frac{c_i}{f^{k_i}}$; since $c_i \in \psi(x) \cap A_{mk_i}$, $\frac{c_i^m}{f^{mk_i}} \in x$ implying that $\frac{c_i}{f^{k_i}} \in x$ and $z \in x$.
  - $x \subseteq \phi(\psi(x))$: if $\frac{a}{f^k} \in x$ for $a \in A_{mk}$, then $a \in \psi(x)$ for $\frac{a_i^m}{f^i} = \frac{a^m}{f^{mk}} = \left(\frac{a}{f^k}\right)^m \in x$ if $i = mk$ or 0 otherwise. Thus $\frac{a}{f^k} \in \operatorname{span}\left\{\frac{c}{1} | c \in g(x)\right\} \cap A_f^0$ since $\frac{a}{f^k} = \frac{a}{1}\frac{1}{f^k}$.
- $\psi \circ \phi = 1$:
  - $\psi(\phi(x)) \subseteq x$: let $z \in \psi(\phi(x))$ and i be a natural number, since $\frac{z_i^m}{f^i} \in \phi(x)$, $\frac{z_i^m}{f^i}$ can be written as $\sum_j \frac{c_j}{f^{n_j}} \frac{d_j}{1}$ with $d_j \in x$, by multiplying a suitable power of f, $z_i^m f^N = \sum_j c_j d_j f^{N_j}$ for some N implying that $z_i \in x$.
  - $x \le \psi(\phi(x))$: if $z \in x$, then $z_i \in x$ for all natural number i by homogeneity. So $\frac{z_i^m}{f^i} = \frac{1}{f^i}\left(\frac{z_i}{1}\right)^m \in \phi(x)$ because $\frac{z_i}{1} \in \phi(x)$.

Thus ϕ and ψ are both bijections implying that ψ is continuous as well: $D(f)$ has a basis of the form $D(f) \cap D(a)$, thus it is sufficient to prove that preimages of these sets are open. By considering $\phi(D(f) \cap D(a)) = \bigcup_i \phi(D(f) \cap D(a_i))$, each $\phi(D(f) \cap D(a_i))$ is open because

$\phi(D(f)\cap D(a_i)) = D\left(\frac{a_i^m}{f^i}\right)$ in SpecA_f^0. To prove $\phi(D(f)\cap D(a_i)) = D\left(\frac{a_i^m}{f^i}\right)$, it is sufficient to prove $\phi^{-1}(D\left(\frac{a_i^m}{f^i}\right)) = D(f)\cap D(a)$ and this proven in continuity of ϕ. Since ϕ and ψ are inverses to each other, preimage of $D(f)\cap D(a)$ is indeed $\phi(D(f)\cap D(a))$. Thus we have proven that $\phi : D(f) \cong \text{Spec}A_f^0$ as topological spaces. ◀

Let ϕ and ψ be the continuous functions defined in the previous proof, U be an open subset of Spec A_f^0, s be a section on $\phi^{-1}(U)$ and $x\in U$, then $\psi(x)\in\phi^{-1}(U)$, hence $s(\psi(x)) = \frac{n}{d}\in A^0_{\psi(x)}$ for some $i\in\mathbb{N}$ and $n,d\in\mathcal{A}_i$. Keeping the same notation, a ring homomorphism $\alpha_U : \phi_*(\mathcal{O}_{\text{Proj}}\,|_{D(f)})(U) \to \mathcal{O}_{\text{Spec}A_f^0}(U)$ can be defined as $s\mapsto \left(x\mapsto \frac{nd^{m-1}/f^i}{d^m/f^i}\right)$ where $n,d\in\mathcal{A}_i$. Assuming that α is indeed a ring homomorphism, it is easy to check that the following diagram commutes whenever $V\subseteq U$:

$$\begin{array}{ccc}
\mathcal{O}_{\text{Proj}|_{D(f)}}(h^{-1}(U)) & \xrightarrow{\phi_U} & \mathcal{O}_{\text{Spec}A_f^0}(U) \\
\downarrow & & \downarrow \\
\mathcal{O}_{\text{Proj}|_{D(f)}}(h^{-1}(V)) & \xrightarrow{\phi_V} & \mathcal{O}_{\text{Spec}A_f^0}(V)
\end{array}.$$

▶ **Lemma 10.** *For any open subset $U\subseteq$ Spec A_f^0, α_U is well-defined; hence α defines a morphism of sheaves.*

Proof. It is clear that both the numerator and denominator have degrees zero. $d^m/f^i\notin x$ follows from $d\notin\psi(x)$.

- α_U preserves one: let $x\in U$, suppose $1=\alpha_U(1)(x)=\frac{n}{d}$ where $n,d\in A_i$, by definition of localisation, for some $c\notin\psi(1)$, $nc=dc$. Since $c\notin\psi(1)$, there is some $j\in\mathbb{N}$, $\frac{c_j^m}{f^j}\notin x$. Then $(nc)_{i+j}=(dc)_{i+j}=nc_j=dc_j$. Hence
$$\frac{c_j^m}{f^j}\frac{nd^{m-1}}{f^i}=\frac{c_j^m nd^{m-1}}{f^{i+j}}=\frac{(dc_j)^m}{f^{i+j}}=\frac{c_j^m}{f^j}\frac{d^m}{f^i},$$
i.e. $\alpha_U(1)(x)=1$ for all x. Similarly, α_U preserves zero as well.
- α_U preserves addition: let s_1,s_2 be two sections and $x\in U$, write $s_1(\psi(x))=\frac{n_1}{d_1}$ with $n_1,d_1\in A_{i_1}$, $s_2(\psi(x))=\frac{n_2}{d_2}$ with $n_2,d_2\in A_{i_2}$ and $(s_1+s_2)(\psi(x))=\frac{n_{12}}{d_{12}}=\frac{n_1}{d_1}+\frac{n_2}{d_2}$ with $n_{12},d_{12}\in A_{i_{12}}$. Thus it is sufficient to check
$$\frac{n_{12}d_{12}^{m-1}/f^{i_{12}}}{d_{12}^m/f^{i_{12}}}=\frac{n_1d_1^{m-1}/f^{i_1}}{d_1^m/f^{i_1}}+\frac{n_2d_2^{m-1}/f^{i_2}}{d_2^m/f^{i_2}}.$$
From $\frac{n_{12}}{d_{12}}=\frac{n_1}{d_1}+\frac{n_2}{d_2}=\frac{n_1d_2+n_2d_1}{d_1d_2}$, so we can find a $c\notin\psi(x)$, such that
$$n_{12}d_1d_2c=(n_1d_2+n_2d_1)d_{12}c.$$
Since $c\notin\psi(x)$, there is some $j\in\mathbb{N}$ such that $\frac{c_j^m}{f^j}\notin x$. Then by taking the $i_1{+}i_2{+}i_{12}{+}j$-th projection $n_{12}d_1d_2c_j=(n_1d_2+n_2d_1)d_{12}c_j$, implying the desired equality by multiplying c_j^m/f^j. Using this, one can check that by multiplying $\frac{c_j^m}{f^j}$, the desired equality can be proved. Similarly, $\alpha_U(s_1s_2)(x)=\alpha_U(s_1)(x)\alpha_U(s_2)(x)$.

Hence, α_U is indeed a ring homomorphism, all that is left to check is that $\alpha_U(s)$ is locally a fraction. Since s is locally quotient, for any $x\in U$, there is some open set $V\subseteq\text{Proj}\mathcal{A}$ such that $\psi(x)\in V\subseteq\phi^{-1}(U)$ such that $s(y)=\frac{a}{b}$ for all $y\in V$ where $a,b\in A_n$ and $b\notin y$, then to check $\alpha_U(s)$ is locally quotient, use the open subset $\phi(V)$ and check that for all $z\in\phi(V)$, $\alpha_U(s)(z)=\frac{ab^{m-1}}{b^m}$. ◀

In the other direction, if $s \in \mathcal{O}_{\mathrm{Spec} A_f^0}(U)$ and $y \in \phi^{-1}(U)$, then $\phi(y) \in U$, so $s(\phi(y))$ can be written as $\frac{a}{b}$ where $a, b \in A_f^0$; then a can be written as $\frac{n_a}{f^{i_a}}$ for some $n_a \in A_{mi_a}$ and b as $\frac{n_b}{f^{i_b}}$ for some $n_b \in A_{mi_b}$. Hence, a ring homomorphism $\beta_U : \mathcal{O}_{\mathrm{Spec} A_f^0}(U) \to \mathcal{O}_{\mathrm{Proj}}|_{D(f)}\,(\phi^{-1}(U))$ can be defined as $s \mapsto \left(y \mapsto \frac{n_a f_b^i}{n_b f^{i_a}}\right)$. Assuming β is well defined, it is easy to check that the assignment $U \mapsto \beta_U$ is natural so that β is a natural transformation.

▶ **Lemma 11.** *For any open subset $U \subseteq \mathrm{Spec}\, A_f^0$, β_U is well-defined; hence β defines a morphism of sheaves.*

Proof. Since s locally is a fraction, there are open sets $\phi(y) \in V \subseteq U$, such that for all $z \in V$, $s(z)$ is $\frac{a/f^{l_1}}{b/f^{l_2}}$. Then on $\phi^{-1}(V) \subseteq \phi^{-1}(U)$, $\psi_U(s)(y)$ is always $\frac{af^{l_2}}{bf^{l_1}}$. Then it is sufficient to check that β_U is a ring homomorphism:

- β_U preserves one. Suppose $\frac{1}{1} = 1(\phi(y)) = \frac{a}{b}$ where $a, b \in A_f^0$, then by definition of localisation, there is some $\frac{c}{f^l} \notin \phi(y)$ such that
$$\frac{n_a c}{f^{i_a+l}} = \frac{n_b c}{f^{i_b+l}};$$
by definition of localisation again, there is some $n_1 \in \mathbb{N}$ such that
$$n_a c f^{i_b+l+n_1} = n_b c f^{i_a+l+n_1}.$$
Hence $\beta_U(1)(y) = \frac{n_a f^{b_i}}{n_b f^{i_a}} = 1$. Similarly, β_U preserves zero as well.
- β_U preserves multiplication, let s_1, s_2 be two sections and $y \in \phi^{-1}(U)$, by writing $s_1(\phi(y))$ as $\frac{a_1/f^{i_1}}{b_1/f^{j_1}}$, $s_2(\phi(y))$ as $\frac{a_2/f^{i_2}}{b_2/f^{i_2}}$ and $(s_1 s_2)(\phi(y)) = \frac{a_{12}/f^{i_{12}}}{b_{12}/f^{j_{12}}}$, one deduces that $\frac{a_1 a_2/f^{i_1+i_2}}{b_1 b_2/f^{j_1+j_2}} = \frac{a_{12}/f^{i_{12}}}{b_{12}/f^{j_{12}}}$, by definition of localisation, it implies that, there is some $\frac{c}{f^l}$ such that
$$\frac{a_1 a_2 b_{12} c}{f^{i_1+i_2+j_{12}+l}} = \frac{a_{12} b_1 b_2 c}{f^{i_{12}+j_1+j_2+l}}.$$
Hence, there is some $L \in \mathbb{N}$, such that
$$a_1 a_2 b_{12} c f^{i_{12}+j_1+j_2+l+L} = a_{12} b_1 b_2 c f^{i_1+i_2+j_{12}+l+L},$$
implying that
$$\frac{a_{12} f^{j_{12}}}{b_{12} f^{i_{12}}} = \frac{a_1 f^{j_1}}{b_1 f^{i_1}} \cdot \frac{a_2 f^{j_2}}{b_2 f^{i_2}}.$$
Similarly, β_U preserves addition as well.

◀

▶ **Theorem 12.** *$\phi_*(\mathcal{O}_{\mathrm{Proj}\,\mathcal{A}}|_{D(f)})$ and $\mathcal{O}_{\mathrm{Spec}\,A_f^0}$ are isomorphic as sheaves.*

Proof. By combining Lemma 10 and Lemma 11, it is sufficient to check $\alpha \circ \beta$ and $\beta \circ \alpha$ are both identities.

- $\beta \circ \alpha = 1$: let $s \in \mathcal{O}_{\mathrm{Proj}}|_{D(f)}\,(\phi^{-1}(U))$, then for $x \in \phi^{-1}(U)$
$$\alpha_U(s) = x \mapsto \frac{nd^{m-1}/f^i}{d^m/f^i},$$
where $s(x) = \frac{n}{d}$. Thus, by definition
$$\beta_U(\alpha_U(s))(x) = \frac{nd^{m-1} f^i}{d^m f^i} = \frac{n}{d} = s(x).$$

- $\alpha \circ \beta = 1$: let $s \in \mathcal{O}_{\mathrm{Spec} A_f^0}(U)$, then for $x \in U$

$$\beta_U(s) = x \mapsto \frac{n_a f^{i_b}}{n_b f^{i_a}}$$

where $s(x) = \frac{n_a/f^{i_a}}{n_b/f^{i_b}}$. Thus

$$\phi_U(\psi_U(s))(x) = \frac{n_a f^{i_b}(n_b f^{i_a})^{m-1}/f^j}{(n_b f^{i_a})^m/f^j} = \frac{n_a/f^{i_a}}{n_b/f^{i_b}} = s(x).$$

◀

▶ **Corollary 13.** $(\mathrm{Proj}\,\mathcal{A}, \mathcal{O}_{\mathrm{Proj}\,\mathcal{A}})$ *is a scheme.*

3 Formalisation details

3.1 Homogeneous Ideal

Let A be an R-algebra and an ι-grading $\mathcal{A} : \iota \to R$-submodules of A [16], `ideal.is_homogeneous` is the proposition of an ideal being homogeneous and `homogeneous_ideal` is the type of all homogeneous ideals of A. Note that, by this implementation, homogeneous ideals are not literally ideals, for this reason, $\mathrm{Proj}\,\mathcal{A}$ cannot be implemented as a subset of $\mathrm{Spec}\,A$.

```
def ideal.is_homogeneous : Prop :=
∀ (i : ι) ⦃r : A⦄, r ∈ I → (direct_sum.decompose 𝒜 r i : A) ∈ I

structure homogeneous_ideal extends submodule A A :=
(is_homogeneous' : ideal.is_homogeneous 𝒜 to_submodule)

def homogeneous_ideal.to_ideal (I : homogeneous_ideal 𝒜) : ideal A :=
    I.to_submodule

lemma homogeneous_ideal.is_homogeneous (I : homogeneous_ideal 𝒜) :
  I.to_ideal.is_homogeneous 𝒜 := I.is_homogeneous'

def homogeneous_ideal.irrelevant : homogeneous_ideal 𝒜 :=
⟨(graded_ring.proj_zero_ring_hom 𝒜).ker, sorry⟩
```

3.2 Homogeneous Localisation

If x is a multiplicatively closed subset, then the homogeneous localisation of A at x is defined to be the subring of localised ring A_x consisting of elements of degree zero. This ring is implemented as triples $\{(i, a, b) : \iota \times \mathcal{A}_i \times \mathcal{A}_i \mid b \notin x\}$ under the equivalence relation that $(i_1, a_1, b_1) \approx (i_2, a_2, b_2) \overset{\text{def}}{\iff} \frac{a_1}{b_1} = \frac{a_2}{b_2}$ in A_x. This approach gives an induction principle, though the construction still uses classical reasoning, many lemmas will be automatic because of rich APIs in mathlib about quotient spaces already; compared to the subring approach, one would need to write corresponding lemmas manually by excessively invoking `classical.some` and `classical.some_spec`.

```
variables {ι R A: Type*} [add_comm_monoid ι] [decidable_eq ι]
variables [comm_ring R] [comm_ring A] [algebra R A]
variables (𝒜 : ι → submodule R A) [graded_algebra 𝒜]
variables (x : submonoid A)
```

```
structure num_denom_same_deg :=
(deg : ι) (num denom : 𝒜 deg) (denom_mem : (denom : A) ∈ x)

def embedding (p : num_denom_same_deg 𝒜 x) : localization x :=
localization.mk p.num ⟨p.denom, p.denom_mem⟩

def homogeneous_localization : Type* :=
quotient (setoid.ker $ embedding 𝒜 x)
```

Then if `(y : homogeneous_localization 𝒜 x)`, its value, degree, numerator and denominator can all be defined by using induction/recursion principles for quotient spaces:

```
variable (y : homogeneous_localization 𝒜 x)

def val : localization x :=
  quotient.lift_on' y (num_denom_same_deg.embedding 𝒜 x) $ λ _ _, id

def num : A := (quotient.out' y).num
def denom : A := (quotient.out' y).denom
def deg : ι := (quotient.out' y).deg

lemma denom_mem : y.denom ∈ x := (quotient.out' y).denom_mem
lemma num_mem_deg : y.num ∈ 𝒜 f.deg := (quotient.out' y).num.2
lemma denom_mem_deg : y.denom ∈ 𝒜 y.deg := (quotient.out' y).denom.2
lemma eq_num_div_denom :
 y.val = localization.mk y.num ⟨y.denom, y.denom_mem⟩ := sorry
```

3.3 Zariski Topology

In this section A will be graded by $\mathbb{N}$. $\operatorname{Proj}\mathcal{A}$ is formalised a structure:

```
structure projective_spectrum :=
(as_homogeneous_ideal : homogeneous_ideal 𝒜)
(is_prime : as_homogeneous_ideal.to_ideal.is_prime)
(not_irrelevant_le : ¬(homogeneous_ideal.irrelevant 𝒜 ≤ as_homogeneous_ideal))
```

After building more APIs around `projective_spectrum`, Zariski topology with a basis of basic open sets can be formalised as:

```
def zero_locus (s : set A) : set (projective_spectrum 𝒜) :=
{x | s ⊆ x.as_homogeneous_ideal}

instance zariski_topology : topological_space (projective_spectrum 𝒜) :=
topological_space.of_closed (set.range (zero_locus 𝒜)) sorry sorry sorry

def basic_open (r : A) : topological_space.opens (projective_spectrum 𝒜) :=
{ val := { x | r ∉ x.as_homogeneous_ideal },
  property := ⟨{r}, set.ext $ λ x, set.singleton_subset_iff.trans $ not_not.symm⟩ }

lemma is_topological_basis_basic_opens : topological_space.is_topological_basis
  (set.range (λ (r : A), (basic_open 𝒜 r : set (projective_spectrum 𝒜)))) :=
sorry
```

3.4 Locally Ringed Space

mathlib provides `Top.presheaf.is_sheaf_iff_is_sheaf_comp` to check sheaf condition by composing a forgetful functor and `Top.subsheaf_to_Types` to construct subsheaf of types satisfying a local predicate [6]; $\mathcal{O}_{\mathrm{Spec}}$ in mathlib adopted this approach [5], and structure sheaf of Proj will also be constructed in this way. `is_locally_fraction` is a local predicate expressing "being locally a homogeneous fraction" in Section 2.3:

```
def is_fraction_prelocal : prelocal_predicate (λ (x : Proj 𝒜),  A⁰ₓ ) :=
{ pred := λ U f, is_fraction f,
  res := by rintros V U i f ⟨j, r, s, w⟩; exact ⟨j, r, s, λ y, w (i y)⟩ }

def is_locally_fraction : local_predicate (λ (x : Proj 𝒜),  A⁰ₓ ) :=
(is_fraction_prelocal 𝒜).sheafify

def structure_sheaf_in_Type : sheaf Type* (Proj 𝒜):=
subsheaf_to_Types (is_locally_fraction 𝒜)
```

The presheaf of rings is also defined as `structure_presheaf_in_CommRing` and checked that composition with forgetful functor is naturally isomorphic to the (underlying presheaf) of `structure_sheaf_in_Type` which implies that `structure_presheaf_in_CommRing` satisfies the sheaf condition as well by using `Top.presheaf.is_sheaf_iff_is_sheaf_comp`.

```
def structure_presheaf_in_CommRing : presheaf CommRing (Proj 𝒜) :=
{ obj := λ U, CommRing.of ((structure_sheaf_in_Type 𝒜).1.obj U), ..sorry }

def structure_presheaf_comp_forget :
  structure_presheaf_in_CommRing 𝒜 ⋙ (forget CommRing) ≅
  (structure_sheaf_in_Type 𝒜).1 := sorry

def Proj.structure_sheaf : sheaf CommRing (Proj 𝒜) :=
⟨structure_presheaf_in_CommRing 𝒜,
  (is_sheaf_iff_is_sheaf_comp _ _).mpr
    (is_sheaf_of_iso (structure_presheaf_comp_forget 𝒜).symm
    (structure_sheaf_in_Type 𝒜).cond)⟩
```

Then following Theorem 7, `stalk_to_fiber_ring_hom` is a family of ring homomorphism $\prod_x \mathcal{O}_{\mathrm{Proj}\,\mathcal{A},x} \to A^0_x$ obtained by universal property of colimit with its right inverse as a family of function `homogeneous_localization_to_stalk`:

```
def stalk_to_fiber_ring_hom (x : Proj 𝒜) :
  (Proj.structure_sheaf 𝒜).presheaf.stalk x ⟶ CommRing.of  A⁰ₓ  :=
limits.colimit.desc (((open_nhds.inclusion x).op) ⋙ (Proj.structure_sheaf 𝒜).1)
  sorry

def section_in_basic_open (x : Proj 𝒜) :
  Π (f :  A⁰ₓ ),
    (Proj.structure_sheaf 𝒜).1.obj (op (Proj.basic_open 𝒜 f.denom)) :=
λ f, ⟨λ y, quotient.mk' ⟨_, ⟨f.num, _⟩, ⟨f.denom,_⟩, _⟩, _⟩

def homogeneous_localization_to_stalk (x : Proj 𝒜) :
   A⁰ₓ  → (Proj.structure_sheaf 𝒜).presheaf.stalk x :=
λ f, (Proj.structure_sheaf 𝒜).presheaf.germ
  (⟨x, homogeneous_localization.mem_basic_open _ x f⟩ : Proj.basic_open _ f.denom)
  (section_in_basic_open _ x f)
```

```
def Proj.stalk_iso' (x : Proj 𝒜) :
  (Proj.structure_sheaf 𝒜).presheaf.stalk x ≃+* CommRing.of  A⁰ₓ  :=
ring_equiv.of_bijective (stalk_to_fiber_ring_hom _ x)
  ⟨sorry, function.surjective_iff_has_right_inverse.mpr
    ⟨homogeneous_localization_to_stalk 𝒜 x, sorry⟩⟩
```

Hence establishing that Proj $\mathcal{A}$ is a locally ringed space:

```
def Proj.to_LocallyRingedSpace : LocallyRingedSpace :=
{ local_ring := λ x, @@ring_equiv.local_ring _
    (show local_ring  A⁰ₓ , from infer_instance) _
    (Proj.stalk_iso' 𝒜 x).symm,
    ..(Proj.to_SheafedSpace 𝒜) }
```

3.5 Affine cover

```
variables {f : A} {m : ℕ} (f_deg : f ∈ 𝒜 m) (x : Proj| D(f))
```

`Spec.T` and `Proj.T` denotes the topological space associated with each locally ringed spaces. Let $0 < m \in \mathbb{N}$ and $f \in \mathcal{A}_m$ and $x \in D(f)$, by following Theorem 9, the continuous function ϕ and ψ in Section 2.3 is formalised as `Proj_iso_Spec_Top_component.to_Spec` and `Proj_iso_Spec_Top_component.from_Spec` respectively; $\phi \circ \psi = 1$ and $\psi \circ \phi = 1$ are recorded as `Proj_iso_Spec_Top_component.to_Spec_from_Spec` and `.from_Spec_to_Spec` respectively:

```
namespace Proj_iso_Spec_Top_component
namespace to_Spec
def carrier : ideal A⁰_f :=
ideal.comap (algebra_map A⁰_f A_f)
  (ideal.span $ algebra_map A (away f) '' x.val.as_homogeneous_ideal)

def to_fun : Proj.T| D(f) → Spec.T A⁰_f :=
λ x,
⟨carrier 𝒜 x, sorry /-a proof for primeness-/⟩
end to_Spec

def to_Spec (f : A) : Proj.T| D(f) ⟶ Spec.T A_f :=
{ to_fun := to_Spec.to_fun 𝒜 f,
  continuous_to_fun := begin
    apply is_topological_basis.continuous (Spec.is_topological_basis_basic_opens),
    sorry
  end }

namespace from_Spec
def carrier (q : Spec.T A⁰_f) : set A :=
{a | ∀ i, (quotient.mk' ⟨_, ⟨proj 𝒜 i a ^ m, _⟩, ⟨f^i, _⟩, _⟩ : A⁰_f) ∈ q.1}

def carrier.as_ideal : ideal A := { carrier := carrier f_deg q, ..sorry }
def carrier.as_homogeneous_ideal : homogeneous_ideal 𝒜 :=
⟨carrier.as_ideal f_deg hm q, sorry⟩
lemma carrier.relevant :
  ¬homogeneous_ideal.irrelevant 𝒜 ≤ carrier.as_homogeneous_ideal f_deg hm q :=
sorry
lemma carrier.as_ideal.prime : (carrier.as_ideal f_deg hm q).is_prime :=
(carrier.as_ideal.homogeneous f_deg hm q).is_prime_of_homogeneous_mem_or_mem
```

```
    sorry sorry

def to_fun : Spec.T A⁰_f → Proj.T| D(f) :=
λ q, ⟨⟨carrier.as_homogeneous_ideal f_deg hm q, carrier.as_ideal.prime f_deg hm q,
      carrier.relevant f_deg hm q⟩, sorry⟩
end from_Spec

lemma to_Spec_from_Spec : to_Spec.to_fun 𝒜 f (from_Spec.to_fun f_deg hm x) = x :=
sorry
lemma from_Spec_to_Spec : from_Spec.to_fun f_deg hm (to_Spec.to_fun 𝒜 f x) = x :=
sorry

def from_Spec : Spec.T A⁰_f ⟶ Proj.T| D(f) :=
{ to_fun := from_Spec.to_fun f_deg hm,
  continuous_to_fun := begin
    apply is_topological_basis.continuous,
    sorry
  end }
end Proj_iso_Spec_Top_component

def Proj_iso_Spec_Top_component:
  Proj.T| D(f) ≅ Spec.T (A⁰_f) :=
{ hom := Proj_iso_Spec_Top_component.to_Spec 𝒜 f,
  inv := Proj_iso_Spec_Top_component.from_Spec hm f_deg, ..
  sorry /-composition being identity-/ }
```

Then by following Lemma 11, β is formalised as `Proj_iso_Spec_Sheaf_component.from_Spec`:

```
namespace Proj_iso_Spec_Sheaf_component
namespace from_Spec
variables (V : (opens (Spec A⁰_f))ᵒᵖ)
variables (s : (Spec A⁰_f).presheaf.obj V)
variables (y : ((@opens.open_embedding Proj.T D(f)).is_open_map.functor.op.obj
  ((opens.map (Proj_iso_Spec_Top_component hm f_deg).hom).op.obj V)).unop)
-- For type checking purpose, but basically a verbose way of spelling
-- y is in φ⁻¹(V)

-- Corresponding to evaluating a section in Lemma 11.
def data : structure_sheaf.localizations A⁰_f))
  ((Proj_iso_Spec_Top_component hm f_deg).hom ⟨y.1, _⟩) :=
s.1 ⟨_, _⟩

-- s(φ(y)) = a/b, this is a, see Lemma 11.
def data.num : A⁰_f := sorry

-- s(φ(y)) = a/b, this is b, see Lemma 11
def data.denom : A⁰_f := sorry

-- s ↦ (y ↦ n_a f_b^i / n_b f^{ia}), this is n_a f_b^i, see Lemma 11.
def num : A :=
  (data.num _ hm f_deg s y).num * (data.denom _ hm f_deg s y).denom
end Proj_iso_Spec_Sheaf_component

-- s ↦ (y ↦ n_a f_b^i / n_b f^{ia}), this is n_b f^{ia}, see Lemma 11.
def denom : A :=
```

```
  (data.denom _ hm f_deg s y).num * (data.num _ hm f_deg s y).denom

-- s ↦ (y ↦ n_a f_b^i / n_b f^{i_a}), this is n_a f_b^i / n_b f^{i_a}, see Lemma 11.
def bmk : A⁰_y :=
quotient.mk'
{ deg := (data.num _ hm f_deg s y).deg + (data.denom _ hm f_deg s y).deg,
  num := ⟨num hm f_deg s y, _⟩,
  denom := ⟨denom hm f_deg s y, _⟩,
  denom_mem := denom_not_mem hm f_deg s y }

def to_fun.aux : ((Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj|
    D(f)).presheaf).obj V :=
⟨bmk hm f_deg V s, sorry /-being locally a homogeneous fraction-/⟩

def to_fun : (Spec A⁰_f).presheaf.obj V ⟶
  ((Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f)).presheaf).obj V :=
{ to_fun := λ s, to_fun.aux 𝒜 hm f_deg V s,
  ..sorry /-ring homomorphism proofs-/ }

end from_Spec

def from_Spec : (Spec A⁰_f).presheaf ⟶
  (Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f)).presheaf :=
{ app := λ V, from_Spec.to_fun 𝒜 hm f_deg V,
  naturality' := λ _ _ _, by { ext1, simpa } }
end Proj_iso_Spec_Sheaf_component
```

By following Lemma 10, α is formalised as `Proj_iso_Spec_Sheaf_component.to_Spec`:

```
namespace Proj_iso_Spec_Sheaf_component
namespace to_Spec
variable (U : (opens (Spec.T A⁰_f))ᵒᵖ)
variable (s : ((Proj_iso_Spec_Top_component hm f_deg).hom _*
    (Proj| D(f))).presheaf.obj U) -- (ϕ_*(𝒪_Proj|_{D(f)}))(U)

-- evaluating a section, this is s(ψ(y))
def hl (y : unop U) : homogeneous_localization 𝒜 _ :=
s.1 ⟨((Proj_iso_Spec_Top_component hm f_deg).inv y.1).1, _⟩

-- s ↦ (x ↦ (nd^{m-1}/f^i)/(d^m/f^i)) where n, d ∈ 𝒜_i, this is nd^{m-1}/f^i, see Lemma 10.
def num (y : unop U) : A⁰_f :=
quotient.mk'
{ deg := m * (hl hm f_deg s y).deg,
  num := ⟨(hl hm f_deg s y).num * (hl hm f_deg s y).denom ^ m.pred, _⟩,
  denom := ⟨f^(hl hm f_deg s y).deg, _⟩,
  denom_mem := _ }

-- s ↦ (x ↦ (nd^{m-1}/f^i)/(d^m/f^i)) where n, d ∈ 𝒜_i, this is d^m/f^i, see Lemma 10.
def denom (y : unop U) : A⁰_f :=
quotient.mk'
{ deg := m * (hl hm f_deg s y).deg,
  num := ⟨(hl hm f_deg s y).denom ^ m, _⟩,
  denom := ⟨f ^ (hl hm f_deg s y).deg,_⟩,
  denom_mem := _ }
```

```
-- s ↦ (x ↦ nd^{m-1}/f^i/d^m/f^i) where n, d ∈ 𝒜_i, this is nd^{m-1}/f^i/d^m/f^i, see Lemma 10.
def fmk (y : unop U) : (A⁰_f)_y :=
mk (num hm f_deg s y) ⟨denom hm f_deg s y, _⟩

def to_fun :
  ((Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f))).obj U ⟶
  (Spec A⁰_f).presheaf.obj U :=
{ to_fun := λ s, ⟨λ y, fmk hm f_deg s y, sorry /-proof of being locally a
    fraction-/⟩, ..sorry /-proof of being a ring homomorphism-/},
end to_Spec

def to_Spec :
  (Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f)).presheaf ⟶
  (Spec A⁰_f).presheaf :=
{ app := λ U, to_Spec.to_fun hm f_deg U,
  naturality' := λ U V subset1, by { ext1, simpa } }
end Proj_iso_Spec_Sheaf_component
```

Hence it has been shown that $(D(f), \mathcal{O}_{\operatorname{Proj}\mathcal{A}}) \cong (\operatorname{Spec} A_f^0, \mathcal{O}_{\operatorname{Spec} A_f^0})$ as locally ringed spaces and hence $(\operatorname{Proj}\mathcal{A}, \mathcal{O}_{\operatorname{Proj}\mathcal{A}})$ is a scheme.

```
def Sheaf_component:
  (Proj_iso_Spec_Top_component hm f_deg).hom _* (Proj| D(f)).presheaf ≅
  (Spec A⁰_f).presheaf :=
{ hom := Proj_iso_Spec_Sheaf_component.to_Spec 𝒜 hm f_deg,
  inv := Proj_iso_Spec_Sheaf_component.from_Spec 𝒜 hm f_deg,
  ..sorry /-composition is identity-/ }

def iso:
  (Proj| D(f)) ≅ Spec A⁰_f :=
let H : (Proj| D(f)).to_PresheafedSpace ≅ (Spec A⁰_f).to_PresheafedSpace :=
  PresheafedSpace.iso_of_components
    (Proj_iso_Spec_Top_component hm f_deg) (Sheaf_component 𝒜 f_deg hm) in
LocallyRingedSpace.iso_of_SheafedSpace_iso
{ hom := H.1, inv := H.2, hom_inv_id' := H.3, inv_hom_id' := H.4 }

def Proj.to_Scheme : Scheme :=
{ local_affine := sorry,..Proj }
```

4 Conclusion

Though the calculations involving localised ring and localised-localised ring are not sophisticated to perform with a pen and some papers, the process is considerably more cumbersome, if not harder in a theorem prover for the following reasons: 1. mathematicians always liberally write "let $\frac{a}{b} \in A_x$" as I did in Section 2, but this is not immediately clear in a theorem prover for every element in A_x is a fraction is a theorem, not a definition and the denominator b carries a proof that $b \in x$ sometimes rendering `rewrite` unusable; 2. equality like `x*y = y*x` in A_x can be proved by `ring` tactic but `a / b * c / d = c / b * a / d` can only be proved manually and this phenomenon is greatly exacerbated when equalities involved are long, similar to these found in Section 2. Originally, I expected propositional equalities that are not equal by definition such as $\phi(\psi(y)) = y$ in Theorem 9 will pose a challenge, but the difficulty is less severe: indeed, I only need to prove some redundant lemma like $\phi(\psi(y))$

is in some open sets that clearly contains y; the reason is that in this project I did not compare algebraic structures depending on propositional equality, i.e. $\mathcal{O}_y$ and $\mathcal{O}_{\phi(\psi(y))}$; but foreseeably, this difficulty will come back when one starts to develop the theory of projective variety furtherer.

Since a large part of modern algebraic geometry depends on Proj construction, much potential future research is possible: calculating cohomology of projective spaces; proving $\mathrm{Proj}\,\mathcal{A}$ is not affine; defining projective morphisms; Serre's twisting sheaves to name a few. Other approaches to Proj construction also exists, for example, by gluing a family of schemes together; however, since there is no other formalisation of Proj construction, I could not compare different approaches or compare capabilities of formalising modern algebraic geometry of different theorem provers. Thus I would like to conclude this paper with an invitation/challenge — state and formalise something involving more than affine schemes in your preferred theorem prover; for the only way to know which, if any, theorem provers handles modern mathematics satisfactorily is to actually formalise more modern mathematics.

References

1 Anthony Bordg, Lawrence Paulson, and Wenda Li. Simple type theory is not too simple: Grothendieck's schemes without dependent types. *Experimental Mathematics*, 31(2):364–382, 2022. arXiv:https://doi.org/10.1080/10586458.2022.2062073, doi:10.1080/10586458.2022.2062073.

2 Kevin Buzzard, Chris Hughes, Kenny Lau, Amelia Livingston, Ramon Fernández Mir, and Scott Morrison. Schemes in lean. *Experimental Mathematics*, 31(2):355–363, 2022.

3 Laurent Chicli. *Une formalisation des faisceaux et des schémas affines en théorie des types avec Coq.* PhD thesis, INRIA, 2001.

4 Wei-Liang Chow. On compact complex analytic varieties. *American Journal of Mathematics*, 71(4):893–914, 1949. URL: http://www.jstor.org/stable/2372375.

5 Mathlib Contributors. Lean mathlib. https://github.com/leanprover-community/mathlib, 2022.

6 Mathlib Contributors. Mathlib documentation. https://leanprover-community.github.io/mathlib_docs, 2023.

7 Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris Van Doorn, and Jakob von Raumer. The lean theorem prover (system description). In *International Conference on Automated Deduction*, pages 378–388. Springer, 2015.

8 Roger Godement. Topologie algébrique et théorie des faisceaux. *Publications de*, 1, 1958.

9 Robin Hartshorne. Graduate texts in mathematics. *Algebraic Geometry*, 52, 1977.

10 Saunders MacLane and Ieke Moerdijk. *Sheaves in geometry and logic: A first introduction to topos theory.* Springer Science & Business Media, 2012.

11 Anders Mörtberg and Max Zeuner. Towards a formalization of affine schemes in cubical agda.

12 Amnon Neeman. *Algebraic and analytic geometry.* Number 345. Cambridge University Press, 2007.

13 Jean-Pierre Serre. Géométrie analytique et géométrie algébrique. *Ann. Inst. Fourier, VI (1955–56)*, pages 1–42, 1955.

14 The Stacks Project Authors. *Stacks Project.* https://stacks.math.columbia.edu, 2018.

15 Ravi Vakil. The rising sea: Foundations of algebraic geometry. 2017. *URL http://virtualmath1.stanford. edu/˜ vakil/216blog*, 24:29.

16 Eric Wieser and Jujian Zhang. Graded rings in lean's dependent type theory. In *International Conference on Intelligent Computer Mathematics*, pages 122–137. Springer, 2022.